Optimizing LRGV Water Allocation

Start stuff

```
Off[General::spell];
Off[General::spell1];
```

Set-Up

We begin by establishing demand, cost, and loss parameters for all sectors. Water units throughout are 1000s of af. Prices and costs and values are per 1000af, except in the very last results table. We solve for unknown "A_i" coefficients using presumed elasticities and known demand points. p represents price at the expansion point, which may or may not be the same as C which represents constant mc of processing for the forthcoming opt analysis. Five sectors are present. ALL data should be entered in the **pm** matrix.

```
lbl1 = {{"Sector", "Ai", "ϵi", "qi", "pi", "Li", "Ci"}};
pm = {{"muni-sm",   0., -.32, 40.0, 440000., 0.2, 440000.},
     {"muni-lg",   0., -.32, 100.0, 400000., 0.1, 400000.},
     {"ag-field",  0., -.7, 547.5, 20000., 0.20, 20000.},
     {"ag-veget",  0., -.4, 371.9, 24000., 0.20, 24000.},
     {"ag-citrus", 0., -.4, 227.5, 22000., 0.20, 20000.}};
Do[
  pm[[i, 2]] = pm[[i, 4]] * (1. - pm[[i, 6]]) / (pm[[i, 5]]^pm[[i, 3]]),
  {i, 1, 5}
 ];
lbled1 = Join[lbl1, pm];
TableForm[lbled1]
```

Sector	A_i	ϵ_i	q_i	p_i	L_i	C_i
muni-sm	2046.7	-0.32	40.	440000.	0.2	440000.
muni-lg	5583.44	-0.32	100.	400000.	0.1	400000.
ag-field	448947.	-0.7	547.5	20000.	0.2	20000.
ag-veget	16811.3	-0.4	371.9	24000.	0.2	24000.
ag-citrus	9932.09	-0.4	227.5	22000.	0.2	20000.

Compute Demand, Total Benefit, and Marginal Net Benefit Functions

Some intermediate functionals are computed here for verification, later plotting, or later optimization. The q_i function is simple demand for processed water by sector i. The remaining functions all relate to raw water. p_i is inverse demand and is expressed as a function of Q rather than Q_i for ease of later plotting. Due to nonconvergence of the integral under the demand function (demand is asymptotic to price axis), it is not possible to wholy express total benefits as a function of Q. Only the TB and TC columns are used is the forthcoming optimization work.

```
lbl2 = {{"Sector", "qi", "pi", "TB(Qi)-Trunc(i)", "TC(Qi)", "MNB(Qi)"}};
unkn = {{Q1, Q2, Q3, Q4, Q5}};
funct = Table[0., {i, 5}, {j, 6}];
Do[
    {funct[[i, 1]] = pm[[i, 1]],
    funct[[i, 2]] = pm[[i, 2]] * p^pm[[i, 3]],
    funct[[i, 3]] = ((1. - pm[[i, 6]]) * Q /
                  pm[[i, 2]]) ^ (1. / pm[[i, 3]]),
    funct[[i, 4]] = pm[[i, 2]] ^ (-1. / pm[[i, 3]]) * (pm[[i, 3]] / (pm[[i, 3]] + 1.)) *
     (unkn[[1, i]] * (1. - pm[[i, 6]])) ^ ((pm[[i, 3]] + 1.) / pm[[i, 3]]),
    funct[[i, 5]] = pm[[i, 7]] * unkn[[1, i]] * (1. - pm[[i, 6]]),
    funct[[i, 6]] = D[funct[[i, 4]], unkn[[1, i]]] - pm[[i, 7]] * (1. - pm[[i, 6]])
    }, {i, 1, 5}
  ];
lbled2 = Join[lbl2, funct];
TableForm[lbled2]
```

Sector	q_i	p_i	TB(Q_i)-Trunc(i)	TC(Q_i)	MNB(Q_i)
muni-sm	$\frac{2046.7}{p^{0.32}}$	$\frac{4.46571\times10^{10}}{Q^{3.125}}$	$-\frac{1.68121\times10^{10}}{Q1^{2.125}}$	352 000. Q1	$-352\,000. + \frac{3.57256\times10^{10}}{Q1^{3.125}}$
muni-lg	$\frac{5583.44}{p^{0.32}}$	$\frac{7.11312\times10^{11}}{Q^{3.125}}$	$-\frac{3.01261\times10^{11}}{Q2^{2.125}}$	360 000. Q2	$-360\,000. + \frac{6.40181\times10^{11}}{Q2^{3.125}}$
ag-field	$\frac{448\,947.}{p^{0.7}}$	$\frac{1.63308\times10^{8}}{Q^{1.42857}}$	$-\frac{3.04841\times10^{8}}{Q3^{0.428571}}$	16 000. Q3	$-16\,000. + \frac{1.30646\times10^{8}}{Q3^{1.42857}}$
ag-veget	$\frac{16\,811.3}{p^{0.4}}$	$\frac{6.40143\times10^{10}}{Q^{2.5}}$	$-\frac{3.41409\times10^{10}}{Q4^{1.5}}$	19 200. Q4	$-19\,200. + \frac{5.12114\times10^{10}}{Q4^{2.5}}$
ag-citrus	$\frac{9932.09}{p^{0.4}}$	$\frac{1.71742\times10^{10}}{Q^{2.5}}$	$-\frac{9.15957\times10^{9}}{Q5^{1.5}}$	16 000. Q5	$-16\,000. + \frac{1.37393\times10^{10}}{Q5^{2.5}}$

Optimize (using Mathematica 6)

The outputed table for this section is designed to correspond with *WRR* Table 3. Here we have to ignore the resultant values provided for the objective function. They are of opposite sign AND do not include majority of total benefits due to nonconvergence of integral under Cobb-Douglas demands. The omission of a portion of total benefits does not invalidate the optimization results because it is fixed and unrelated to selected Q's. Note the error that begins to accumulate as some sectors become satiated; compare first and last columns when available water is 1300 thousand acre-feet.

```
obj = funct[[1, 4]] - funct[[1, 5]] + funct[[2, 4]] - funct[[2, 5]] + funct[[3, 4]] -
   funct[[3, 5]] + funct[[4, 4]] - funct[[4, 5]] + funct[[5, 4]] - funct[[5, 5]];
results = {{"Avail Q ", " MNB", "Q1-sm", "Q2-lg", "Q3-fld", "Q4-veg", "Q5-cit", "summd Q"}};
Do[

  {answ = NMaximize[{obj, Q1 + Q2 + Q3 + Q4 + Q5 ≤ availQ && Q1 ≥ 0 && Q2 ≥ 0 && Q3 ≥ 0 && Q4 ≥ 0 && Q5 ≥ 0},
     {Q1, Q2, Q3, Q4, Q5}],
    answ2 = { {availQ, funct[[1, 6]] / 1000., Q1, Q2, Q3, Q4, Q5, Q1 + Q2 + Q3 + Q4 + Q5}} /.
     Part[answ, 2],
 results = Join[results, answ2]},
    {availQ, 800., 1300., 100.}

 ];
TableForm[results]
```

Avail Q	MNB	Q1-sm	Q2-lg	Q3-fld	Q4-veg	Q5-cit	summd Q
800.	31.0859	38.9313	97.3845	257.184	253.028	153.472	800.
900.	19.8482	39.304	98.2973	311.273	279.967	171.159	900.
1000.	12.2374	39.5649	98.936	367.867	305.329	188.303	1000.
1100.	6.84515	39.7542	99.3991	426.69	329.197	204.96	1100.
1200.	2.88425	39.8957	99.745	487.527	351.652	221.18	1200.
1300.	2.83812×10^{-7}	40.	100.	547.5	371.9	236.341	1295.74

Draw

■ Demands

```
Plot[{funct[[2, 3]], funct[[3, 3]]}, {Q, 0, 1500},
         PlotRange -> {0, 10 000}]
```



■ Marginal Net Benefits

```
Block[
    {Q1 = Q,
        Q2 = Q,
        Q3 = Q,
        Q4 = Q,
        Q5 = Q},
        Plot[{funct[[1, 6]], funct[[2, 6]],
                  funct[[3, 6]], funct[[4, 6]],
                  funct[[5, 6]]},
              {Q, 0, 600},
            PlotRange -> {0, 40000}]
    ]
```